
;*****************************************************
;Programa de envio e recepção de dados via serial
;Função: Este programa faz eco serial: tudo que chega, retorna.
;Obs: Não usa interrupção
;Programador: Rogério M. Branco
;Data: 30/11/2003
;Versão: 1.0
;*****************************************************

        org 0000h
        AJMP INICIO

        org 0100h
INICIO: MOV P1,#00H     ;ZERA A PORTA P1
        CLR EA          ;DESABILITA QUALQUER INTERRUPÇÃO
        ACALL SET_TCS   ;FAZ CHAMADA DE FUNÇÃO PARA AJUSTE DE TC0 E TC1
        ACALL SET_SER   ;FAZ CHAMADA DE FUNÇÃO PARA AJUSTE DE SERIAL
        SETB TR1        ;DISPARO DO TC1 - GERADOR DE TAXA SERIAL
ESPERA: ACALL REC_SER   ;FAZ CHAMA DE FUNÇÃO ESPERA E RECEBE SERIAL
        ACALL ENV_SER   ;ENVIA DADO RECEBIDO PELA SERIAL
        AJMP ESPERA     ;RETORNA PARA ESPERA = AGUARDANDO NOVO BYTE PELA SERIAL

;****************************************************************************
;FUNÇÕES UTILIZADAS
;****************************************************************************

SET_TCS:MOV TMOD,#00100000B     ;TMOD=GATE CT M1 M0 GATE CT M1 M0
                                ;    =  0   0  1  0   0  0   0  0 
                                ;TC1 EM MODO 2 = IRÁ GERAR TAXA SERIAL
        RET

SET_SER:MOV SCON,#01010000B     ;SCON=SM0 SM1 SM2 REN TB8 RB8 TI RI     
                                ;    = 0   1   0   1   0   0   0  0
                                ;ATIVA SERIAL MODO 1 E RECEPÇÃO

        ORL PCON,#10000000B     ;PCON=SMOD *   *   *  GF1 GF0 PD IDL
                                ;    = 1   0   0   0   0   0   0  0
                                ;ATIVA DIVISOR POR 16 = DOBRO DA TAXA SERIAL
                                
                                ;CALCULO TX SERIAL:
                                ; 9600 = (K * FCLOCK)/(32*12*(256-TH1)
                                ; TH1= 256 - ((K * FCLOCK)/(32*12*9600))
                                ; TH1= 256 - ((2 * 12M)/(32*12*9600))
                                ; TH1= 256 - 6,5104167
                                ; TH1= 249,48958333 = 249,5 = 250d = 0FAH

                                ; a TX SERIAL REAL SERÁ:
                                ; TX = (K * FCLOCK)/(32*12*(256-TH1)
                                ; TX = (2 * 12M)/(32*12*(256-0FAH)
                                ; TX = 10416,67
                                
                                ; O ERRO SERÁ 10416,67-9600 / 9600 = 8,5% 

                                ; SE TH1= 249,48958333 = 249,5 = 249d = 0F9H

                                ; a TX SERIAL REAL SERÁ:
                                ; TX = (K * FCLOCK)/(32*12*(256-TH1)
                                ; TX = (2 * 12M)/(32*12*(256-0F9H)
                                ; TX = 8928,57

                                ; O ERRO SERÁ 8928,57-9600 / 9600 = 7% 

        MOV TH1,#0F9H
        MOV TL1,#0F9H           ;TH1=TL1 POIS É AUTOCARGA
        RET

REC_SER:JNB RI,REC_SER          ;ESPERA ATÉ QUE CHEGUE ALGO PELA SERIAL
        CLR RI                  ;LIMPA FLAG DE AVISO DE RECEPÇÃO SERIAL

        MOV A,SBUF              ;JOGA PARA a O QUE CHEGOU PELA SERIAL
        RET

ENV_SER:MOV SBUF,A              ;ENVIA DADO PELA SERIAL
        JNB TI,$                ;ESPERA ATÉ QUE SBUF FIQUE VAZIO = BYTE ENVIADO
        CLR TI                  ;LIMPA FLAG DE AVISO DE TRANSMISSÃO SERIAL
        RET
;****************************************************************************
        END

